

# Eigenvectors and Eigenvalues

Worksheet by Russell Blyth and Mike May, S.J.

```
> restart:with(plots):with(LinearAlgebra):  
Warning, the name changecoords has been redefined
```

## Outline

Goals of this worksheet:

- 1) Learn the Maple commands to compute eigenvectors
- 2) Visualize the action of a matrix using eigenvectors and eigenvalues.
- 3) Learn to do the step by step computations for finding eigenvectors.
- 4) See some computations for which eigenvectors simplify problem solutions.

```
>
```

## Using Maple commands to compute eigenvectors

We first look at the Maple commands for computing eigenvalues and eigenvectors. These commands let us work with eigenvectors and eigenvalues before we work through the detailed computations.

We start with a randomly generated 2 by 2 matrix.

```
> A := RandomMatrix(2,2,generator=rand(-5..5));
```

$$A := \begin{bmatrix} -2 & 2 \\ 5 & -3 \end{bmatrix} \quad (2.1)$$

```
> Eigenvectors(A);
```

$$\left[ \begin{array}{c} -\frac{5}{2} + \frac{1}{2}\sqrt{41} \\ -\frac{5}{2} - \frac{1}{2}\sqrt{41} \end{array} \right], \left[ \begin{array}{cc} \frac{2}{-\frac{1}{2} + \frac{1}{2}\sqrt{41}} & \frac{2}{-\frac{1}{2} - \frac{1}{2}\sqrt{41}} \\ 1 & 1 \end{array} \right] \quad (2.2)$$

The output is an ordered pair, where the first entry is a vector of the Eigenvalues, and the second entry is a matrix whose columns are the corresponding eigenvectors.

If we specify that the output should be a list, we get a set of triples, where the first entry of each triple is an eigenvalue, the second entry is its multiplicity as a root of the characteristic polynomial, and the third entry is a basis for the eigenspace corresponding to that eigenvector. We can extract this data if we first place these two triples into a sequence.

```
> evA := Eigenvectors(A, output='list');
```

$$\text{evA} := \left[ \left[ \begin{array}{c} -\frac{5}{2} + \frac{1}{2}\sqrt{41}, 1, \left\{ \begin{array}{c} \frac{2}{-\frac{1}{2} + \frac{1}{2}\sqrt{41}} \\ 1 \end{array} \right\} \right], \left[ \begin{array}{c} -\frac{5}{2} - \frac{1}{2}\sqrt{41}, \end{array} \right] \right] \quad (2.3)$$

$$1, \left\{ \left[ \begin{array}{c} \frac{2}{- \frac{1}{2} - \frac{1}{2} \sqrt{41}} \\ 1 \end{array} \right] \right\}$$

The eigenvalues:

```
> evA[1][1]; evA[2][1];
```

$$\begin{aligned} & -\frac{5}{2} + \frac{1}{2} \sqrt{41} \\ & -\frac{5}{2} - \frac{1}{2} \sqrt{41} \end{aligned} \tag{2.4}$$

The eigenvectors:

```
> [evA[1][3][1], evA[2][3][1]];
```

$$\left[ \left[ \begin{array}{c} \frac{2}{- \frac{1}{2} + \frac{1}{2} \sqrt{41}} \\ 1 \end{array} \right], \left[ \begin{array}{c} \frac{2}{- \frac{1}{2} - \frac{1}{2} \sqrt{41}} \\ 1 \end{array} \right] \right] \tag{2.5}$$

▼ Exercise:

1) Use the following code to generate a symmetric 3 by 3 matrix. Find the Eigenvalues and Eigenvectors.

```
> B := RandomMatrix(3,3,generator=rand(-5..5),
  outpoptions=[shape=symmetric]);
```

```
>
```

```
>
```

## ▼ Visualizing eigenvectors and eigenvalues in $\mathbb{R}^2$

We can use eigenvectors and eigenvalues to give a description of the action of a transformation from  $\mathbb{R}^2$  to  $\mathbb{R}^2$ . We break this into three cases, when the eigenvalues are real and distinct, when the eigenvalues are complex (must be conjugate in this case), and when the eigenvalues repeat (must be real in this case).

### ▼ Case 1 - Distinct real eigenvalues

To produce distinct real eigenvalues, we start with a matrix that is symmetric

```
> A := RandomMatrix(2,2,generator=rand(-5..5),
  outpoptions=[shape=symmetric]);
```

$$A := \begin{bmatrix} 5 & 2 \\ 2 & 4 \end{bmatrix} \tag{3.1.1}$$

We use the Eigenvector command to find the eigenvalues and corresponding eigenvectors.


```
> EigenA := [Eigenvectors(A)];
  EigenVecA := [Column(EigenA[2],1), Column(EigenA[2],2)];
```

(3.1.2)

$$\text{EigenVecA} := \left[ \left[ \begin{array}{c} 2 \\ -\frac{1}{2} + \frac{1}{2}\sqrt{17} \\ 1 \end{array} \right], \left[ \begin{array}{c} 2 \\ -\frac{1}{2} - \frac{1}{2}\sqrt{17} \\ 1 \end{array} \right] \right] \quad (3.1.2)$$

We plot the eigenvectors along with their images. The original vectors are graphed with double arrow heads and the images with single sided arrow heads.

```
> v1 := arrow(EigenVecA[1],color=blue, shape=arrow,thickness=5):
v2 := arrow(EigenVecA[2],color=red, shape=arrow,thickness=5):
Av1 := arrow(A.EigenVecA[1],color=navy, shape=harpoon,
thickness=3):
Av2 := arrow(A.EigenVecA[2],color=green, shape=harpoon,
thickness=3):
display({v1, v2, Av1, Av2},scaling=constrained);
```



Notice that the images of the eigenvectors are scalar multiples of the eigenvectors.

▼ Exercise:

2) For the matrix A above, give a basis of  $\mathbb{R}^2$  composed of eigenvectors.

Describe the action of A on this basis.

Explain how A acts on a linear combination of the vectors in this basis.

[>  
[>

## Case 2 - Complex eigenvalues

When the eigenvalues are complex, it means the action of the transformation involves a rotation. In the 2 by 2 case this can be guaranteed by letting the two entries on the main diagonal be close in value and the entries on the off diagonal be of opposite signs.

```
> A := <<1,3> | <-4,2>>;  
EigenA := [Eigenvectors(A)];
```


$$A := \begin{bmatrix} 1 & -4 \\ 3 & 2 \end{bmatrix}$$

$$\text{EigenA} := \left[ \begin{bmatrix} \frac{3}{2} + \frac{1}{2} I\sqrt{47} \\ \frac{3}{2} - \frac{1}{2} I\sqrt{47} \end{bmatrix}, \begin{bmatrix} -\frac{4}{\frac{1}{2} + \frac{1}{2} I\sqrt{47}} & -\frac{4}{\frac{1}{2} + \frac{1}{2} I\sqrt{47}} \\ 1 & 1 \end{bmatrix} \right] \quad (3.2.1)$$

In that case we use the Eigenplot command from the Student[LinearAlgebra] package to visualize what happens.

```
> Student[LinearAlgebra][EigenPlot](A, showunitvectors=true,  
scaling=constrained, axes=normal, numvectors=8);
```

### The Images of Unit Vectors and Eigenvectors



The graph has unit vectors starting at the origin in yellow, with the respective images in green starting at the head of the original vector. In this case we have a counterclockwise rotation along with an expansion.

>

#### Case 3 - Repeated eigenvectors

The third case is repeated real eigenvalues. Sometimes, in this case, Maple does not produce a basis of eigenvectors. We will come back to this case in later sections.

```
> A := <<2,0> | <2,2>>;
Eigenvectors(A);
Eigenvectors(A,output='list');
```


$$A := \begin{bmatrix} 2 & 2 \\ 0 & 2 \end{bmatrix}$$

$$\begin{bmatrix} 2 \\ 2 \end{bmatrix}, \begin{bmatrix} 1 & 0 \\ 0 & 0 \end{bmatrix}$$

$$\left[ \left[ 2, 2, \left\{ \begin{bmatrix} 1 \\ 0 \end{bmatrix} \right\} \right] \right] \quad (3.3.1)$$

```
> Student[LinearAlgebra][EigenPlot](A, showunitvectors=true,
scaling=constrained, axes=normal, numvectors=8);
```

The Images of Unit Vectors and Eigenvectors



```
> Eigenvectors(A, output='list');
```

$$\left[ \left[ 2, 2, \left\{ \begin{bmatrix} 1 \\ 0 \end{bmatrix} \right\} \right] \right] \quad (3.3.2)$$

### Exercise:

3) Randomly generate a 2 by 2 matrix with distinct real eigenvalues. (You may need to try the generation several times.) Using the result of the Eigenvectors command, describe the action of the transformation. Plot a graph that illustrates your description.

```
[>
```

4) Randomly generate a 2 by 2 matrix with complex eigenvalues. (You may need to try the generation several times.) Using the result of the Student[LinearAlgebra][Eigenplot] command, describe the action of the transformation.

```
[>
```

## Visualizing eigenvectors and eigenvalues in $\mathbb{R}^3$

### Case 1 - Distinct real eigenvalues

We can repeat the process for 3 by 3 matrices, but we need some computational trickery. Since solving cubics involves using complex numbers in a way that does not easily simplify, we evaluate the decimal approximation of the matrix of eigenvectors. The imaginary part is small enough to be roundoff error, so we eliminate it before graphing.

```
> A := RandomMatrix(3,3,generator=rand(-5..5),
  outputoptions=[shape=symmetric]);
EigenA := [Eigenvectors(A)];
EigenA1 := evalf(EigenA[2]);
EigenA2 := map(Re,evalf(EigenA[2]));
EigenVecA := ([Column(EigenA2,1),Column(EigenA2,2),Column
(EigenA2,3)]):
v1 := arrow(EigenVecA[1],color=blue, shape=arrow,thickness=3)
:
v2 := arrow(EigenVecA[2],color=red, shape=arrow,thickness=3):
v3 := arrow(EigenVecA[3],color=brown, shape=arrow,thickness=
3):
Av1 := arrow(A.EigenVecA[1],color=navy, shape=harpoon,
thickness=3):
Av2 := arrow(A.EigenVecA[2],color=green, shape=harpoon,
thickness=3):
Av3 := arrow(A.EigenVecA[3],color=yellow, shape=harpoon,
thickness=3):
display({v1, v2,v3, Av1, Av2,Av3},scaling=constrained, axes=
normal);
```

```
>
```

► *Exercise:*

```
>
```

### ► Case 2 - Complex eigenvalues

## Computing eigenvalues and eigenvectors - Step by step

While it is useful to have a simple Maple command for Eigenvalues and Eigenvectors, we would also like to step through the computations, so that we can do this by hand if needed.

First define a matrix and the identity of the same size.

```
> A := <<-1,0,5>|<0,4,0>|<5,0,-1>>;
I3 := IdentityMatrix(3);
```

$$A := \begin{bmatrix} -1 & 0 & 5 \\ 0 & 4 & 0 \\ 5 & 0 & -1 \end{bmatrix}$$
$$I3 := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

(5.1)

Compute the characteristic polynomial, which is  $\det(A - xI_3)$

```
> chpol := Determinant(A - x*I3);  
chpol := -96 + 32 x + 2 x2 - x3 (5.2)
```

Set equal to zero and solve.

```
> AEvals := [solve(chpol = 0, x)];  
AEvals := [-6, 4, 4] (5.3)
```

Compute the basis of the first eigenspace by finding the basis of the corresponding nullspace.

```
> AEbasis1 := NullSpace(A-AEvals[1]*I3);  
AEbasis1 :=  $\left\{ \begin{bmatrix} -1 \\ 0 \\ 1 \end{bmatrix} \right\}$  (5.4)
```

Repeat for the second eigenvalue

```
> AEbasis2 := NullSpace(A-AEvals[2]*I3);  
AEbasis2 :=  $\left\{ \begin{bmatrix} 1 \\ 0 \\ 1 \end{bmatrix}, \begin{bmatrix} 0 \\ 1 \\ 0 \end{bmatrix} \right\}$  (5.5)
```

Check that these eigenspace basis vectors are in fact eigenvectors of A

```
> [AEbasis1[1], AEvals[1], A . AEbasis1[1]];  
[AEbasis2[1], AEvals[2], A . AEbasis2[1]];  
[AEbasis2[2], AEvals[3], A . AEbasis2[2]];  
 $\begin{bmatrix} -1 \\ 0 \\ 1 \end{bmatrix}, -6, \begin{bmatrix} 6 \\ 0 \\ -6 \end{bmatrix}$   
 $\begin{bmatrix} 1 \\ 0 \\ 1 \end{bmatrix}, 4, \begin{bmatrix} 4 \\ 0 \\ 4 \end{bmatrix}$   
 $\begin{bmatrix} 0 \\ 1 \\ 0 \end{bmatrix}, 4, \begin{bmatrix} 0 \\ 4 \\ 0 \end{bmatrix}$  (5.6)
```

Note that these vectors are the correct multiples of the eigenvectors.

#### Exercises

(6) Follow the recipe above to find the eigenvalues and eigenvectors of the matrix B.

```
> B := Matrix(3,3,[[0,1,1],[0,2,0],[-2,1,3]]);
```

(7) Repeat for the matrix C.

```
> C := Matrix(3,3,[[4,-3,1],[4,-1,0],[1,7,-4]]);
```

When you are done: what can you say about the matrix C?

► Using eigenvalues and eigenvectors to ease computation